

MARS

REGISTERS  
INSTRUCTION SET  
DIRECTIVES  
SYSCALLS

# Registers

---

MIPS has 32 integer registers.

The hardware architecture specifies that:

- General purpose register \$0 always returns a value of 0.
- General purpose register \$31 is used as the link register for jump and link instructions.
- HI and LO are used to access the multiplier/divider results, accessed by the mfhi (move from high) and mflo commands.

These are the only hardware restrictions on the usage of the general purpose registers.

The various MIPS tool-chains implement specific calling conventions that further restrict how the registers are used. These calling conventions are totally maintained by the tool-chain software and are not required by the hardware.

Register	Number	Usage
zero	0	Constant 0
at	1	Reserved for assembler
v0	2	Used for return values from function calls.
v1	3	
a0	4	Used to pass arguments to procedures and functions.
a1	5	
a2	6	
a3	7	
t0	8	Temporary (Caller-saved, need not be saved by called procedure)
t1	9	
t2	10	
t3	11	
t4	12	
t5	13	
t6	14	
t7	15	
s0	16	Saved temporary (Callee-saved, called procedure must save and restore)
s1	17	
s2	18	
s3	19	
s4	20	
s5	21	
s6	22	
s7	23	
t8	24	Temporary (Caller-saved, need not be saved by called procedure)
t9	25	

k0	26	Reserved for OS kernel
k1	27	
gp	28	Pointer to global area
sp	29	Stack pointer
fp	30	Frame pointer
ra	31	Return address for function calls.

MIPS has 32 floating-point registers (\$f0 – \$f31). Two registers are paired for double precision numbers. Odd numbered registers cannot be used for arithmetic or branching, just as part of a double precision register pair.

# Instruction Set

---

The following list provides a description of basic MIPS instructions. For more information on pseudo-instructions available in MARS, refer to MARS help section.

## Operand Key for Example Instructions

label, target	any textual label
\$t1, \$t2, \$t3	any integer register
\$f2, \$f4, \$f6	even-numbered floating point register
\$f0, \$f1, \$f2	any floating point register
\$8	any Coprocessor 0 register
1	condition flag (0 to 7)
10	unsigned 5-bit integer (0 to 31)
-100	signed 16-bit integer (-32768 to 32767)
100000	signed 32-bit integer (-2147483648 to 2147483647)

## Load & Store addressing mode

-100(\$t2)	sign-extended 16-bit integer added to contents of \$t2
------------	--

## Basic Instructions

<b>abs.d</b> \$f2, \$f4	Floating point absolute value double precision : Set \$f2 to absolute value of \$f4, double precision
<b>abs.s</b> \$f0, \$f1	Floating point absolute value single precision : Set \$f0 to absolute value of \$f1, single precision
<b>add</b> \$t1, \$t2, \$t3	Addition with overflow : set \$t1 to (\$t2 plus \$t3)
<b>add.d</b> \$f2, \$f4, \$f6	Floating point addition double precision : Set \$f2 to double-precision floating point value of \$f4 plus \$f6
<b>add.s</b> \$f0, \$f1, \$f3	Floating point addition single precision : Set \$f0 to single-precision floating point value of \$f1 plus \$f3
<b>addi</b> \$t1, \$t2, -100	Addition immediate with overflow : set \$t1 to (\$t2 plus signed 16-bit immediate)
<b>addiu</b> \$t1, \$t2, -100	Addition immediate unsigned without overflow : set \$t1 to (\$t2 plus signed 16-bit immediate), no overflow
<b>addu</b> \$t1, \$t2, \$t3	Addition unsigned without overflow : set \$t1 to (\$t2 plus \$t3), no overflow
<b>and</b> \$t1, \$t2, \$t3	Bitwise AND : Set \$t1 to bitwise AND of \$t2 and \$t3
<b>andi</b> \$t1, \$t2, 100	Bitwise AND immediate : Set \$t1 to bitwise AND of \$t2 and zero-extended 16-bit immediate
<b>bc1f</b> 1, label	Branch if specified FP condition flag false (BC1F, not BCLF) : If Coprocessor 1 condition flag specified by immediate is false (zero) then branch to statement at label's address
<b>bc1f</b> label	Branch if FP condition flag 0 false (BC1F, not BCLF) : If Coprocessor 1 condition flag 0 is false (zero) then branch to statement at label's address

<b>bc1t</b> <i>1, label</i>	Branch if specified FP condition flag true (BC1T, not BCLT) : If Coprocessor 1 condition flag specified by immediate is true (one) then branch to statement at label's address
<b>bc1t</b> <i>label</i>	Branch if FP condition flag 0 true (BC1T, not BCLT) : If Coprocessor 1 condition flag 0 is true (one) then branch to statement at label's address
<b>beq</b> <i>\$t1, \$t2, label</i>	Branch if equal : Branch to statement at label's address if \$t1 and \$t2 are equal
<b>bgez</b> <i>\$t1, label</i>	Branch if greater than or equal to zero : Branch to statement at label's address if \$t1 is greater than or equal to zero
<b>bgezal</b> <i>\$t1, label</i>	Branch if greater than or equal to zero and link : If \$t1 is greater than or equal to zero, then set \$ra to the Program Counter and branch to statement at label's address
<b>bgtz</b> <i>\$t1, label</i>	Branch if greater than zero : Branch to statement at label's address if \$t1 is greater than zero
<b>blez</b> <i>\$t1, label</i>	Branch if less than or equal to zero : Branch to statement at label's address if \$t1 is less than or equal to zero
<b>bltz</b> <i>\$t1, label</i>	Branch if less than zero : Branch to statement at label's address if \$t1 is less than zero
<b>bltzal</b> <i>\$t1, label</i>	Branch if less than zero and link : If \$t1 is less than or equal to zero, then set \$ra to the Program Counter and branch to statement at label's address
<b>bne</b> <i>\$t1, \$t2, label</i>	Branch if not equal : Branch to statement at label's address if \$t1 and \$t2 are not equal
<b>break</b>	Break execution : Terminate program execution with exception
<b>break</b> <i>100</i>	Break execution with code : Terminate program execution with specified exception code
<b>c.eq.d</b> <i>\$f2, \$f4</i>	Compare equal double precision : If \$f2 is equal to \$f4 (double-precision), set Coprocessor 1 condition flag 0 true else set it false
<b>c.eq.d</b> <i>1, \$f2, \$f4</i>	Compare equal double precision : If \$f2 is equal to \$f4 (double-precision), set Coprocessor 1 condition flag specified by immediate to true else set it to false
<b>c.eq.s</b> <i>\$f0, \$f1</i>	Compare equal single precision : If \$f0 is equal to \$f1, set Coprocessor 1 condition flag 0 true else set it false
<b>c.eq.s</b> <i>1, \$f0, \$f1</i>	Compare equal single precision : If \$f0 is equal to \$f1, set Coprocessor 1 condition flag specified by immediate to true else set it to false
<b>c.le.d</b> <i>\$f2, \$f4</i>	Compare less or equal double precision : If \$f2 is less than or equal to \$f4 (double-precision), set Coprocessor 1 condition flag 0 true else set it false
<b>c.le.d</b> <i>1, \$f2, \$f4</i>	Compare less or equal double precision : If \$f2 is less than or equal to \$f4 (double-precision), set Coprocessor 1 condition flag specified by immediate true else set it false

<b>c.le.s</b> \$f0,\$f1	Compare less or equal single precision : If \$f0 is less than or equal to \$f1, set Coprocessor 1 condition flag 0 true else set it false
<b>c.le.s</b> 1,\$f0,\$f1	Compare less or equal single precision : If \$f0 is less than or equal to \$f1, set Coprocessor 1 condition flag specified by immediate to true else set it to false
<b>c.lt.d</b> \$f2,\$f4	Compare less than double precision : If \$f2 is less than \$f4 (double-precision), set Coprocessor 1 condition flag 0 true else set it false
<b>c.lt.d</b> 1,\$f2,\$f4	Compare less than double precision : If \$f2 is less than \$f4 (double-precision), set Coprocessor 1 condition flag specified by immediate to true else set it to false
<b>c.lt.s</b> \$f0,\$f1	Compare less than single precision : If \$f0 is less than \$f1, set Coprocessor 1 condition flag 0 true else set it false
<b>c.lt.s</b> 1,\$f0,\$f1	Compare less than single precision : If \$f0 is less than \$f1, set Coprocessor 1 condition flag specified by immediate to true else set it to false
<b>ceil.w.d</b> \$f1,\$f2	Ceiling double precision to word : Set \$f1 to 32-bit integer ceiling of double-precision float in \$f2
<b>ceil.w.s</b> \$f0,\$f1	Ceiling single precision to word : Set \$f0 to 32-bit integer ceiling of single-precision float in \$f1
<b>clo</b> \$t1,\$t2	Count number of leading ones : Set \$t1 to the count of leading one bits in \$t2 starting at most significant bit position
<b>clz</b> \$t1,\$t2	Count number of leading zeroes : Set \$t1 to the count of leading zero bits in \$t2 starting at most significant bit position
<b>cvt.d.s</b> \$f2,\$f1	Convert from single precision to double precision : Set \$f2 to double precision equivalent of single precision value in \$f1
<b>cvt.d.w</b> \$f2,\$f1	Convert from word to double precision : Set \$f2 to double precision equivalent of 32-bit integer value in \$f1
<b>cvt.s.d</b> \$f1,\$f2	Convert from double precision to single precision : Set \$f1 to single precision equivalent of double precision value in \$f2
<b>cvt.s.w</b> \$f0,\$f1	Convert from word to single precision : Set \$f0 to single precision equivalent of 32-bit integer value in \$f1
<b>cvt.w.d</b> \$f1,\$f2	Convert from double precision to word : Set \$f1 to 32-bit integer equivalent of double precision value in \$f2
<b>cvt.w.s</b> \$f0,\$f1	Convert from single precision to word : Set \$f0 to 32-bit integer equivalent of single precision value in \$f1
<b>div</b> \$t1,\$t2	Division with overflow : Divide \$t1 by \$t2 then set LO to quotient and HI to remainder (use mfhi to access HI, mflo to access LO)
<b>div.d</b> \$f2,\$f4,\$f6	Floating point division double precision : Set \$f2 to double-precision floating point value of \$f4 divided by \$f6
<b>div.s</b> \$f0,\$f1,\$f3	Floating point division single precision : Set \$f0 to single-precision floating point value of \$f1 divided by \$f3

<b>divu \$t1,\$t2</b>	Division unsigned without overflow : Divide unsigned \$t1 by \$t2 then set LO to quotient and HI to remainder (use mfhi to access HI, mflo to access LO)
<b>eret</b>	Exception return : Set Program Counter to Coprocessor 0 EPC register value, set Coprocessor Status register bit 1 (exception level) to zero
<b>floor.w.d \$f1,\$f2</b>	Floor double precision to word : Set \$f1 to 32-bit integer floor of double-precision float in \$f2
<b>floor.w.s \$f0,\$f1</b>	Floor single precision to word : Set \$f0 to 32-bit integer floor of single-precision float in \$f1
<b>j target</b>	Jump unconditionally : Jump to statement at target address
<b>jal target</b>	Jump and link : Set \$ra to Program Counter (return address) then jump to statement at target address
<b>jalr \$t1</b>	Jump and link register : Set \$ra to Program Counter (return address) then jump to statement whose address is in \$t1
<b>jalr \$t1,\$t2</b>	Jump and link register : Set \$t1 to Program Counter (return address) then jump to statement whose address is in \$t2
<b>jr \$t1</b>	Jump register unconditionally : Jump to statement whose address is in \$t1
<b>lb \$t1,-100(\$t2)</b>	Load byte : Set \$t1 to sign-extended 8-bit value from effective memory byte address
<b>lbu \$t1,-100(\$t2)</b>	Load byte unsigned : Set \$t1 to zero-extended 8-bit value from effective memory byte address
<b>ldc1 \$f2,-100(\$t2)</b>	Load double word Coprocessor 1 (FPU) : Set \$f2 to 64-bit value from effective memory doubleword address
<b>lh \$t1,-100(\$t2)</b>	Load halfword : Set \$t1 to sign-extended 16-bit value from effective memory halfword address
<b>lhu \$t1,-100(\$t2)</b>	Load halfword unsigned : Set \$t1 to zero-extended 16-bit value from effective memory halfword address
<b>ll \$t1,-100(\$t2)</b>	Load linked : Paired with Store Conditional (sc) to perform atomic read-modify-write. Treated as equivalent to Load Word (lw) because MARS does not simulate multiple processors.
<b>lui \$t1,100</b>	Load upper immediate : Set high-order 16 bits of \$t1 to 16-bit immediate and low-order 16 bits to 0
<b>lw \$t1,-100(\$t2)</b>	Load word : Set \$t1 to contents of effective memory word address
<b>lwc1 \$f1,-100(\$t2)</b>	Load word into Coprocessor 1 (FPU) : Set \$f1 to 32-bit value from effective memory word address
<b>lwl \$t1,-100(\$t2)</b>	Load word left : Load from 1 to 4 bytes left-justified into \$t1, starting with effective memory byte address and continuing through the low-order byte of its word
<b>lwr \$t1,-100(\$t2)</b>	Load word right : Load from 1 to 4 bytes right-justified into \$t1, starting with effective memory byte address and continuing through the high-order byte of its word

<b>madd \$t1,\$t2</b>	Multiply add : Multiply \$t1 by \$t2 then increment HI by high-order 32 bits of product, increment LO by low-order 32 bits of product (use mfhi to access HI, mflo to access LO)
<b>maddu \$t1,\$t2</b>	Multiply add unsigned : Multiply \$t1 by \$t2 then increment HI by high-order 32 bits of product, increment LO by low-order 32 bits of product, unsigned (use mfhi to access HI, mflo to access LO)
<b>mfc0 \$t1,\$8</b>	Move from Coprocessor 0 : Set \$t1 to the value stored in Coprocessor 0 register \$8
<b>mfc1 \$t1,\$f1</b>	Move from Coprocessor 1 (FPU) : Set \$t1 to value in Coprocessor 1 register \$f1
<b>mfhi \$t1</b>	Move from HI register : Set \$t1 to contents of HI (see multiply and divide operations)
<b>mflo \$t1</b>	Move from LO register : Set \$t1 to contents of LO (see multiply and divide operations)
<b>mov.d \$f2,\$f4</b>	Move floating point double precision : Set double precision \$f2 to double precision value in \$f4
<b>mov.s \$f0,\$f1</b>	Move floating point single precision : Set single precision \$f0 to single precision value in \$f1
<b>movf \$t1,\$t2</b>	Move if FP condition flag 0 false : Set \$t1 to \$t2 if FPU (Coprocessor 1) condition flag 0 is false (zero)
<b>movf \$t1,\$t2,1</b>	Move if specified FP condition flag false : Set \$t1 to \$t2 if FPU (Coprocessor 1) condition flag specified by the immediate is false (zero)
<b>movf.d \$f2,\$f4</b>	Move floating point double precision : If condition flag 0 false, set double precision \$f2 to double precision value in \$f4
<b>movf.d \$f2,\$f4,1</b>	Move floating point double precision : If condition flag specified by immediate is false, set double precision \$f2 to double precision value in \$f4
<b>movf.s \$f0,\$f1</b>	Move floating point single precision : If condition flag 0 is false, set single precision \$f0 to single precision value in \$f1
<b>movf.s \$f0,\$f1,1</b>	Move floating point single precision : If condition flag specified by immediate is false, set single precision \$f0 to single precision value in \$f1e
<b>movn \$t1,\$t2,\$t3</b>	Move conditional not zero : Set \$t1 to \$t2 if \$t3 is not zero
<b>movn.d \$f2,\$f4,\$t3</b>	Move floating point double precision : If \$t3 is not zero, set double precision \$f2 to double precision value in \$f4
<b>movn.s \$f0,\$f1,\$t3</b>	Move floating point single precision : If \$t3 is not zero, set single precision \$f0 to single precision value in \$f1
<b>movt \$t1,\$t2</b>	Move if FP condition flag 0 true : Set \$t1 to \$t2 if FPU (Coprocessor 1) condition flag 0 is true (one)
<b>movt \$t1,\$t2,1</b>	Move if specified FP condition flag true : Set \$t1 to \$t2 if FPU (Coprocessor 1) condition flag specified by the immediate is true (one)
<b>movt.d \$f2,\$f4</b>	Move floating point double precision : If condition flag 0 true, set double precision \$f2 to double precision value in \$f4

<b>movt.d</b> \$f2,\$f4,1	Move floating point double precision : If condition flag specified by immediate is true, set double precision \$f2 to double precision value in \$f4e
<b>movt.s</b> \$f0,\$f1	Move floating point single precision : If condition flag 0 is true, set single precision \$f0 to single precision value in \$f1e
<b>movt.s</b> \$f0,\$f1,1	Move floating point single precision : If condition flag specified by immediate is true, set single precision \$f0 to single precision value in \$f1e
<b>movz</b> \$t1,\$t2,\$t3	Move conditional zero : Set \$t1 to \$t2 if \$t3 is zero
<b>movz.d</b> \$f2,\$f4,\$t3	Move floating point double precision : If \$t3 is zero, set double precision \$f2 to double precision value in \$f4
<b>movz.s</b> \$f0,\$f1,\$t3	Move floating point single precision : If \$t3 is zero, set single precision \$f0 to single precision value in \$f1
<b>msub</b> \$t1,\$t2	Multiply subtract : Multiply \$t1 by \$t2 then decrement HI by high-order 32 bits of product, decrement LO by low-order 32 bits of product (use mfhi to access HI, mflo to access LO)
<b>msubu</b> \$t1,\$t2	Multiply subtract unsigned : Multiply \$t1 by \$t2 then decrement HI by high-order 32 bits of product, decrement LO by low-order 32 bits of product, unsigned (use mfhi to access HI, mflo to access LO)
<b>mtc0</b> \$t1,\$8	Move to Coprocessor 0 : Set Coprocessor 0 register \$8 to value stored in \$t1
<b>mtc1</b> \$t1,\$f1	Move to Coprocessor 1 (FPU) : Set Coprocessor 1 register \$f1 to value in \$t1
<b>mthi</b> \$t1	Move to HI register : Set HI to contents of \$t1 (see multiply and divide operations)
<b>mtlo</b> \$t1	Move to LO register : Set LO to contents of \$t1 (see multiply and divide operations)
<b>mul</b> \$t1,\$t2,\$t3	Multiplication without overflow : Set HI to high-order 32 bits, LO and \$t1 to low-order 32 bits of the product of \$t1 and \$t2 (use mfhi to access HI, mflo to access LO)
<b>mul.d</b> \$f2,\$f4,\$f6	Floating point multiplication double precision : Set \$f2 to double-precision floating point value of \$f4 times \$f6
<b>mul.s</b> \$f0,\$f1,\$f3	Floating point multiplication single precision : Set \$f0 to single-precision floating point value of \$f1 times \$f3
<b>mult</b> \$t1,\$t2	Multiplication : Set hi to high-order 32 bits, lo to low-order 32 bits of the product of \$t1 and \$t2 (use mfhi to access hi, mflo to access lo)
<b>multu</b> \$t1,\$t2	Multiplication unsigned : Set HI to high-order 32 bits, LO to low-order 32 bits of the product of unsigned \$t1 and \$t2 (use mfhi to access HI, mflo to access LO)
<b>neg.d</b> \$f2,\$f4	Floating point negate double precision : Set double precision \$f2 to negation of double precision value in \$f4
<b>neg.s</b> \$f0,\$f1	Floating point negate single precision : Set single precision \$f0 to negation of single precision value in \$f1
<b>nop</b>	Null operation : machine code is all zeroes

<b>nor</b> \$t1,\$t2,\$t3	Bitwise NOR : Set \$t1 to bitwise NOR of \$t2 and \$t3
<b>or</b> \$t1,\$t2,\$t3	Bitwise OR : Set \$t1 to bitwise OR of \$t2 and \$t3
<b>ori</b> \$t1,\$t2,100	Bitwise OR immediate : Set \$t1 to bitwise OR of \$t2 and zero-extended 16-bit immediate
<b>round.w.d</b> \$f1,\$f2	Round double precision to word : Set \$f1 to 32-bit integer round of double-precision float in \$f2
<b>round.w.s</b> \$f0,\$f1	Round single precision to word : Set \$f0 to 32-bit integer round of single-precision float in \$f1
<b>sb</b> \$t1,-100(\$t2)	Store byte : Store the low-order 8 bits of \$t1 into the effective memory byte address
<b>sc</b> \$t1,-100(\$t2)	Store conditional : Paired with Load Linked (ll) to perform atomic read-modify-write. Stores \$t1 value into effective address, then sets \$t1 to 1 for success. Always succeeds because MARS does not simulate multiple processors.
<b>sdcl</b> \$f2,-100(\$t2)	Store double word from Coprocessor 1 (FPU) : Store 64 bit value in \$f2 to effective memory doubleword address
<b>sh</b> \$t1,-100(\$t2)	Store halfword : Store the low-order 16 bits of \$t1 into the effective memory halfword address
<b>sll</b> \$t1,\$t2,10	Shift left logical : Set \$t1 to result of shifting \$t2 left by number of bits specified by immediate
<b>sllv</b> \$t1,\$t2,\$t3	Shift left logical variable : Set \$t1 to result of shifting \$t2 left by number of bits specified by value in low-order 5 bits of \$t3
<b>slt</b> \$t1,\$t2,\$t3	Set less than : If \$t2 is less than \$t3, then set \$t1 to 1 else set \$t1 to 0
<b>slti</b> \$t1,\$t2,-100	Set less than immediate : If \$t2 is less than sign-extended 16-bit immediate, then set \$t1 to 1 else set \$t1 to 0
<b>sltiu</b> \$t1,\$t2,-100	Set less than immediate unsigned : If \$t2 is less than sign-extended 16-bit immediate using unsigned comparison, then set \$t1 to 1 else set \$t1 to 0
<b>sltu</b> \$t1,\$t2,\$t3	Set less than unsigned : If \$t2 is less than \$t3 using unsigned comparison, then set \$t1 to 1 else set \$t1 to 0
<b>sqrtd</b> \$f2,\$f4	Square root double precision : Set \$f2 to double-precision floating point square root of \$f4
<b>sqrts</b> \$f0,\$f1	Square root single precision : Set \$f0 to single-precision floating point square root of \$f1
<b>sra</b> \$t1,\$t2,10	Shift right arithmetic : Set \$t1 to result of sign-extended shifting \$t2 right by number of bits specified by immediate
<b>srav</b> \$t1,\$t2,\$t3	Shift right arithmetic variable : Set \$t1 to result of sign-extended shifting \$t2 right by number of bits specified by value in low-order 5 bits of \$t3
<b>srl</b> \$t1,\$t2,10	Shift right logical : Set \$t1 to result of shifting \$t2 right by number of bits specified by immediate
<b>srlv</b> \$t1,\$t2,\$t3	Shift right logical variable : Set \$t1 to result of shifting \$t2 right by number of bits specified by value in low-order 5 bits of \$t3
<b>sub</b> \$t1,\$t2,\$t3	Subtraction with overflow : set \$t1 to (\$t2 minus \$t3)

<b>sub.d</b> \$f2,\$f4,\$f6	Floating point subtraction double precision : Set \$f2 to double-precision floating point value of \$f4 minus \$f6
<b>sub.s</b> \$f0,\$f1,\$f3	Floating point subtraction single precision : Set \$f0 to single-precision floating point value of \$f1 minus \$f3
<b>subu</b> \$t1,\$t2,\$t3	Subtraction unsigned without overflow : set \$t1 to (\$t2 minus \$t3), no overflow
<b>sw</b> \$t1,-100(\$t2)	Store word : Store contents of \$t1 into effective memory word address
<b>swc1</b> \$f1,-100(\$t2)	Store word from Coprocessor 1 (FPU) : Store 32 bit value in \$f1 to effective memory word address
<b>swl</b> \$t1,-100(\$t2)	Store word left : Store high-order 1 to 4 bytes of \$t1 into memory, starting with effective byte address and continuing through the low-order byte of its word
<b>swr</b> \$t1,-100(\$t2)	Store word right : Store low-order 1 to 4 bytes of \$t1 into memory, starting with high-order byte of word containing effective byte address and continuing through that byte address
<b>syscall</b>	Issue a system call : Execute the system call specified by value in \$v0
<b>teq</b> \$t1,\$t2	Trap if equal : Trap if \$t1 is equal to \$t2
<b>teqi</b> \$t1,-100	Trap if equal to immediate : Trap if \$t1 is equal to sign-extended 16 bit immediate
<b>tge</b> \$t1,\$t2	Trap if greater or equal : Trap if \$t1 is greater than or equal to \$t2
<b>tgei</b> \$t1,-100	Trap if greater than or equal to immediate : Trap if \$t1 greater than or equal to sign-extended 16 bit immediate
<b>tgeiu</b> \$t1,-100	Trap if greater or equal to immediate unsigned : Trap if \$t1 greater than or equal to sign-extended 16 bit immediate, unsigned comparison
<b>tgeu</b> \$t1,\$t2	Trap if greater or equal unsigned : Trap if \$t1 is greater than or equal to \$t2 using unsigned comparison
<b>slt</b> \$t1,\$t2	Trap if less than: Trap if \$t1 less than \$t2
<b>slti</b> \$t1,-100	Trap if less than immediate : Trap if \$t1 less than sign-extended 16-bit immediate
<b>sltiu</b> \$t1,-100	Trap if less than immediate unsigned : Trap if \$t1 less than sign-extended 16-bit immediate, unsigned comparison
<b>sltu</b> \$t1,\$t2	Trap if less than unsigned : Trap if \$t1 less than \$t2, unsigned comparison
<b>tne</b> \$t1,\$t2	Trap if not equal : Trap if \$t1 is not equal to \$t2
<b>tnei</b> \$t1,-100	Trap if not equal to immediate : Trap if \$t1 is not equal to sign-extended 16 bit immediate
<b>trunc.w.d</b> \$f1,\$f2	Truncate double precision to word : Set \$f1 to 32-bit integer truncation of double-precision float in \$f2
<b>trunc.w.s</b> \$f0,\$f1	Truncate single precision to word : Set \$f0 to 32-bit integer truncation of single-precision float in \$f1
<b>xor</b> \$t1,\$t2,\$t3	Bitwise XOR (exclusive OR) : Set \$t1 to bitwise XOR of \$t2 and \$t3
<b>xori</b> \$t1,\$t2,100	Bitwise XOR immediate : Set \$t1 to bitwise XOR of \$t2 and zero-extended 16-bit immediate

# Directives

---

The following list provides a description of all directives available in MARS.

Directives	
<b>.align</b>	Align next data item on specified byte boundary (0=byte, 1=half, 2=word, 3=double)
<b>.ascii</b>	Store the string in the Data segment but do not add null terminator
<b>.asciiz</b>	Store the string in the Data segment and add null terminator
<b>.byte</b>	Store the listed value(s) as 8 bit bytes
<b>.data</b>	Subsequent items stored in Data segment at next available address
<b>.double</b>	Store the listed value(s) as double precision floating point
<b>.extern</b>	Declare the listed label and byte length to be a global data field
<b>.float</b>	Store the listed value(s) as single precision floating point
<b>.globl</b>	Declare the listed label(s) as global to enable referencing from other files
<b>.half</b>	Store the listed value(s) as 16 bit halfwords on halfword boundary
<b>.kdata</b>	Subsequent items stored in Kernel Data segment at next available address
<b>.ktext</b>	Subsequent items (instructions) stored in Kernel Text segment at next available address
<b>.set</b>	Set assembler variables. Currently ignored but included for SPIM compatibility
<b>.space</b>	Reserve the next specified number of bytes in Data segment
<b>.text</b>	Subsequent items (instructions) stored in Text segment at next available address
<b>.word</b>	Store the listed value(s) as 32 bit words on word boundary

# Syscall Functions

---

A number of system services, mainly for input and output, are available for use by your MIPS program. They are described in the table below.

## How to use SYSCALL system services

- Step 1. Load the service number in register \$v0.
- Step 2. Load argument values, if any, in \$a0, \$a1, \$a2, or \$f12 as specified.
- Step 3. Issue the SYSCALL instruction.
- Step 4. Retrieve return values, if any, from result registers as specified.

### Example: display the value stored in \$t0 on the console

```
li $v0, 1           # service 1 is print integer
add $a0, $t0, $zero # load desired value into argument register
                    # $a0, using pseudo-op
syscall
```

Service	Code in \$v0	Arguments	Result
print integer	1	\$a0 = integer to print	
print float	2	\$f12 = float to print	
print double	3	\$f12 = double to print	
print string	4	\$a0 = address of null-terminated string to print	
read integer	5		\$v0 contains integer read
read float	6		\$f0 contains float read
read double	7		\$f0 contains double read
read string	8	\$a0 = address of input buffer \$a1 = maximum number of characters to read	<i>See note below table</i>
sbrk (allocate heap memory)	9	\$a0 = number of bytes to allocate	\$v0 contains address of allocated memory

<b>exit (terminate execution)</b>	10		
<b>print character</b>	11	\$a0 = character to print	<i>See note below table</i>
<b>read character</b>	12		\$v0 contains character read
<b>open file</b>	13	\$a0 = address of null-terminated string containing filename \$a1 = flags \$a2 = mode	\$v0 contains file descriptor (negative if error). <i>See note below table</i>
<b>read from file</b>	14	\$a0 = file descriptor \$a1 = address of input buffer \$a2 = maximum number of characters to read	\$v0 contains number of characters read (0 if end-of-file, negative if error). <i>See note below table</i>
<b>write to file</b>	15	\$a0 = file descriptor \$a1 = address of output buffer \$a2 = number of characters to write	\$v0 contains number of characters written (negative if error). <i>See note below table</i>
<b>close file</b>	16	\$a0 = file descriptor	
<b>exit2 (terminate with value)</b>	17	\$a0 = termination result	<i>See note below table</i>
<b>Services 1 through 17 are compatible with the SPIM simulator, other than Open File (13) as described in the Notes below the table. Services 30 and higher are exclusive to MARS.</b>			
<b>time (system time)</b>	30		\$a0 = low order 32 bits of system time \$a1 = high order 32 bits of system time. <i>See note below table</i>
<b>MIDI out</b>	31	\$a0 = pitch (0-127) \$a1 = duration in milliseconds \$a2 = instrument (0-127) \$a3 = volume (0-127)	Generate tone and return immediately. <i>See MARS help</i>
<b>sleep</b>	32	\$a0 = the length of time to sleep in milliseconds.	Causes the MARS Java thread to sleep for (at least) the specified number of milliseconds. This timing will not be precise, as the Java implementation will add some overhead.

<b>MIDI out synchronous</b>	33	\$a0 = pitch (0-127) \$a1 = duration in milliseconds \$a2 = instrument (0-127) \$a3 = volume (0-127)	Generate tone and return upon tone completion. <i>See MARS help</i>
<b>print integer in hexadecimal</b>	34	\$a0 = integer to print	
<b>print integer in binary</b>	35	\$a0 = integer to print	
<b>print integer as unsigned</b>	36	\$a0 = integer to print	
<b>(not used)</b>	37-39		
<b>set seed</b>	40	\$a0 = i.d. of pseudorandom number generator (any int). \$a1 = seed for corresponding pseudorandom number generator.	No values are returned. Sets the seed of the corresponding underlying Java pseudorandom number generator (java.util.Random). <i>See note below table</i>
<b>random int</b>	41	\$a0 = i.d. of pseudorandom number generator (any int).	\$a0 contains the next pseudorandom, uniformly distributed int value from this random number generator's sequence. <i>See note below table</i>
<b>random int range</b>	42	\$a0 = i.d. of pseudorandom number generator (any int). \$a1 = upper bound of range of returned values.	\$a0 contains pseudorandom, uniformly distributed int value in the range 0 = [int] [upper bound], drawn from this random number generator's sequence. <i>See note below table</i>
<b>random float</b>	43	\$a0 = i.d. of pseudorandom number generator (any int).	\$f0 contains the next pseudorandom, uniformly distributed float value in the range 0.0 = f 1.0 from this random number generator's sequence. <i>See note below table</i>
<b>random double</b>	44	\$a0 = i.d. of pseudorandom number generator (any int).	\$f0 contains the next pseudorandom, uniformly distributed double value in the range 0.0 = f 1.0 from this random number generator's sequence. <i>See note below table</i>
<b>(not used)</b>	45-49		

<b>ConfirmDialog</b>	50	\$a0 = address of null-terminated string that is the message to user	\$a0 contains value of user-chosen option 0: Yes 1: No 2: Cancel
<b>InputDialogInt</b>	51	\$a0 = address of null-terminated string that is the message to user	\$a0 contains int read \$a1 contains status value 0: OK status -1: input data cannot be correctly parsed -2: Cancel was chosen -3: OK was chosen but no data had been input into field
<b>InputDialogFloat</b>	52	\$a0 = address of null-terminated string that is the message to user	\$f0 contains float read \$a1 contains status value 0: OK status -1: input data cannot be correctly parsed -2: Cancel was chosen -3: OK was chosen but no data had been input into field
<b>InputDialogDouble</b>	53	\$a0 = address of null-terminated string that is the message to user	\$f0 contains double read \$a1 contains status value 0: OK status -1: input data cannot be correctly parsed -2: Cancel was chosen -3: OK was chosen but no data had been input into field
<b>InputDialogString</b>	54	\$a0 = address of null-terminated string that is the message to user \$a1 = address of input buffer \$a2 = maximum number of characters to read	<i>See Service 8 note below table</i> \$a1 contains status value 0: OK status. Buffer contains the input string. -2: Cancel was chosen. No change to buffer. -3: OK was chosen but no data had been input into field. No change to buffer. -4: length of the input string exceeded the specified maximum. Buffer contains the maximum allowable input string plus a terminating null.
<b>MessageDialog</b>	55	\$a0 = address of null-terminated string that is the message to user \$a1 = the type of message to be displayed: 0: error message, indicated by Error icon	N/A

---

1: information message, indicated by Information icon  
 2: warning message, indicated by Warning icon  
 3: question message, indicated by Question icon  
 other: plain message (no icon displayed)

<b>MessageDialogInt</b>	56	\$a0 = address of null-terminated string that is an information-type message to user \$a1 = int value to display in string form after the first string	N/A
<b>MessageDialogFloat</b>	57	\$a0 = address of null-terminated string that is an information-type message to user \$f12 = float value to display in string form after the first string	N/A
<b>MessageDialogDouble</b>	58	\$a0 = address of null-terminated string that is an information-type message to user \$f12 = double value to display in string form after the first string	N/A
<b>MessageDialogString</b>	59	\$a0 = address of null-terminated string that is an information-type message to user \$a1 = address of null-terminated string to display after the first string	N/A

---

**NOTES: Services numbered 30 and higher are not provided by SPIM**

**Service 8** - Follows semantics of UNIX 'fgets'. For specified length n, string can be no longer than n-1. If it is less than that, adds newline to end. In either case, then pads with null byte.

**Service 11** - Prints ASCII character corresponding to contents of low-order byte.

**Service 13** - MARS implements three flag values: 0 for read-only, 1 for write-only with create, and 9 for write-only with create and append. It ignores mode. The returned file descriptor will be negative if the operation failed. The underlying file I/O implementation uses java.io.FileInputStream.read() to read and java.io.FileOutputStream.write() to write. MARS maintains file descriptors internally and allocates them starting with 0.

**Services 13,14,15** - In MARS 3.7, the result register was changed to \$v0 for SPIM compatibility. It was previously \$a0 as erroneously printed in Appendix B of *Computer Organization and Design*.

**Service 17** - If the MIPS program is run under control of the MARS graphical interface (GUI), the exit code in \$a0 is ignored.

**Service 30** - System time comes from java.util.Date.getTime() as milliseconds since 1 January 1970.

**Services 40-44** use underlying Java pseudorandom number generators provided by the java.util.Random class. Each stream (identified by \$a0 contents) is modeled by a different Random object. There are no default seed values, so use the Set Seed service (40) if replicated random sequences are desired.

**Example of File I/O**

The sample MIPS program below will open a new file for writing, write text to it from a memory buffer, then close it. The file will be created in the directory in which MARS was run.

```
# Sample MIPS program that writes to a new file.
#   by Kenneth Vollmar and Pete Sanderson
        .data
fout:   .asciiz "testout.txt"      # filename for output
buffer: .asciiz "The quick brown fox jumps over the lazy dog."
        .text
#####
# Open (for writing) a file that does not exist
li     $v0, 13      # system call for open file
la     $a0, fout    # output file name
li     $a1, 1       # Open for writing (flags are 0: read, 1: write)
li     $a2, 0       # mode is ignored
syscall      # open a file (file descriptor returned in $v0)
move   $s6, $v0    # save the file descriptor
#####
# Write to file just opened
li     $v0, 15      # system call for write to file
move   $a0, $s6    # file descriptor
la     $a1, buffer  # address of buffer from which to write
li     $a2, 44      # hardcoded buffer length
syscall      # write to file
#####
# Close the file
li     $v0, 16      # system call for close file
move   $a0, $s6    # file descriptor to close
syscall      # close file
```